// SPDX-License-Identifier: MIT
pragma solidity ^0.8.15;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _setOwner(_msgSender());
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
        _;
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _setOwner(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _setOwner(newOwner);
    }

    function _setOwner(address newOwner) private {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

contract OwnableMulticall is Ownable {
    struct Call {
        address target;
        bytes callData;
    }

    constructor(address initOwner) {
        transferOwnership(initOwner);
    }

    //allows contract owner to make a single call and get its return data
    function singleCall(Call memory callInfo) external onlyOwner returns (bytes memory) {
        (bool success, bytes memory callReturnData) = callInfo.target.call(callInfo.callData);
        require(success, "call failed");
        return callReturnData;
    }
    //allows contract owner to make a single call. ignores return data
    function singleCallNoReturn(Call memory callInfo) external onlyOwner {
        (bool success, ) = callInfo.target.call(callInfo.callData);
        require(success, "call failed");
    }
    //allows contract owner to make any number of calls and get their return data
    function multicall(Call[] memory calls) external onlyOwner returns (bytes[] memory returnDataArray) {
        returnDataArray = new bytes[](calls.length);
        for(uint256 i = 0; i < calls.length; i++) {
            (bool success, bytes memory callReturnData) = calls[i].target.call(calls[i].callData);
            require(success, "call failed");
            returnDataArray[i] = callReturnData;
        }
    }
    //allows contract owner to make any number of calls. ignores return data
    function multicallNoReturn(Call[] memory calls) external onlyOwner {
        for(uint256 i = 0; i < calls.length; i++) {
            (bool success, ) = calls[i].target.call(calls[i].callData);
            require(success, "call failed");
        }
    }
    function withdraw() external onlyOwner {
        payable(_msgSender()).transfer(address(this).balance);
    }
    receive() external payable {}
    fallback() external payable {}
}